
Az elvégzett mérési feladatok
LED céllövölde potméterrel – Virányi Ágnes:
A baloldali LED oszlopon gyulladjanak fel véletlenszerűen a LED-ek. A felgyulladások közötti időt (n*100 ms) a potenciométer állásától függően lehessen állítani 16 lépésben. A nyomógombon el kell találni, hogy melyik LED világít. A megnyomás pillanatában a LED-ek állapota kimerevedik, a baloldali LED oszlopon pedig az  a LED világít, maleyikhez tartozó gombot megnyomtuk. Találat esetén a találatot jelző jobboldali LED 4 Hz frekvenciával villog. Az INT megnyomásával újraindítható a mérés. A pergésmentesítést nem kellett megvalósítani. 
A program folyamatábrája:[image: A description...]
Megjegyzések: A véletlenszerűen felvillanó LEDek előállításához a potenciométer legalsó két bitjét használtam, (ADCL regiszter 7-6. bitje) és az AVR specifikációjából kiindulva ezek a bitek elég zajosak, így véletlenszerűen változik az értékük. A beolvasás után a két bit értékétől függően négy állapotot különböztettem meg, és ennek feleltettem meg a bal oldali LED-eket. 
A tesztelésnél itt is problémát jelentett, hogy nem az elvi frekvencián működött a műszer, és így nem voltak helyesek a 4 Hz-es villogás számításai. 

A program assembly forráskódja:

;*************************************************************** 
;* Feladat: LED céllövölde potméterrel
;* Rövid leírás: A baloldali LED oszlopon véletlenszerűen felgyulladó ;*ledeket kell a nyomógombokkal eltalálni
;				 Találat esetén világít a jobboldali LED oszlop megfelelő LEDje, egyébként folytatódik tovább 
;				 a játék.
;				 A potencióméterrel változtathatjuk a játék gyorsaságát.
; 
;* Szerzők: 	Virányi Ágnes
;* Mérőcsoport: CDU2
;
;***************************************************************
;* "AVR ExperimentBoard" port assignment information:
;***************************************************************
;*
;* LED0(P):PortC.0          LED4(P):PortC.4
;* LED1(P):PortC.1          LED5(P):PortC.5
;* LED2(S):PortC.2          LED6(S):PortC.6
;* LED3(Z):PortC.3          LED7(Z):PortC.7        INT:PortE.4
;*
;* SW0:PortG.0     SW1:PortG.1     SW2:PortG.4     SW3:PortG.3
;* 
;* BT0:PortE.5     BT1:PortE.6     BT2:PortE.7     BT3:PortB.7
;*
;***************************************************************
;*
;* AIN:PortF.0     NTK:PortF.1    OPTO:PortF.2     PPINE 5 1OT:PortF.3
;*
;***************************************************************
;*
;* LCD1(VSS) = GND         LCD9(DB2): -
;* LCD2(VDD) = VCC         LCD10(DB3): -
;* LCD3(VO ) = GND         LCD11(DB4): PortA.4
;* LCD4(RS ) = PortA.0     LCD12(DB5): PortA.5
;* LCD5(R/W) = GND         LCD13(DB6): PortA.6
;* LCD6(E  ) = PortA.1     LCD14(007): PortA.7
;* LCD7(DB0) = -           LCD15(BLA): VCC
;* LCD8(DB1) = -           LCD16(BLK): PortB.5 (1=Backlight ON)
;*
;***************************************************************

.include "m128def.inc" ; Definition file for ATmega128 
;* Program Constants 
.equ const =$00 ; Generic Constant Structure example  
;* Program Variables Definitions 

.def temp =r16 ; Temporary Register example 
.def btn = r17 ; gombokat tarolo regiszter
.def btn1 = r18; BTNek kiolvasasahoz segedreg
.def intreg = r19; az int gomb lenyomasakor allitjuk, =1, ha lenyomtak az INTet
.def veletlen = r20; a veletlenszeru LEDek erteket tartalmazza 
.def tmp =r21	; segedregiszter
.def szazad = r22; TIMER idoziteshez hasznalt valtozo
.def led = r23	; a bal LEDsor ertekeit tartalmazza 
.def jobbled= r24; a jobb LEDsor ertekeit tartalmazza 


;*************************************************************** 
;* Reset & Interrupt Vectors  
.cseg 
.org $0000 ; Define start of Code segment 
	jmp RESET ; Reset Handler, jmp is 2 word instruction 
	jmp DUMMY_IT	; Ext. INT0 Handler
	jmp DUMMY_IT	; Ext. INT1 Handler
	jmp DUMMY_IT	; Ext. INT2 Handler
	jmp DUMMY_IT	; Ext. INT3 Handler
	jmp INT_IT	    ; Ext. INT4 Handler (INT gomb)
	jmp DUMMY_IT  	; Ext. INT5 Handler
	jmp DUMMY_IT	; Ext. INT6 Handler
	jmp DUMMY_IT    ; Ext. INT7 Handler
	jmp DUMMY_IT	; Timer2 Compare Match Handler 
	jmp DUMMY_IT	; Timer2 Overflow Handler 
	jmp DUMMY_IT	; Timer1 Capture Event Handler 
	jmp DUMMY_IT	; Timer1 Compare Match A Handler 
	jmp DUMMY_IT	; Timer1 Compare Match B Handler 
	jmp DUMMY_IT	; Timer1 Overflow Handler - Ledek vilogtatasa talat eseten
	jmp TIMER_IT	; Timer0 Compare Match Handler 
	jmp DUMMY_IT	; Timer0 Overflow Handler 
	jmp DUMMY_IT	; SPI Transfer Complete Handler 
	jmp DUMMY_IT	; USART0 RX Complete Handler 
	jmp DUMMY_IT	; USART0 Data Register Empty Hanlder 
	jmp DUMMY_IT	; USART0 TX Complete Handler 
	jmp DUMMY_IT	; ADC Conversion Complete Handler 
	jmp DUMMY_IT	; EEPROM Ready Hanlder 
	jmp DUMMY_IT	; Analog Comparator Handler 
	jmp DUMMY_IT	; Timer1 Compare Match C Handler 
	jmp DUMMY_IT	; Timer3 Capture Event Handler 
	jmp DUMMY_IT	; Timer3 Compare Match A Handler 
	jmp DUMMY_IT	; Timer3 Compare Match B Handler 
	jmp DUMMY_IT	; Timer3 Compare Match C Handler 
	jmp DUMMY_IT	; Timer3 Overflow Handler 
	jmp DUMMY_IT	; USART1 RX Complete Handler 
	jmp DUMMY_IT	; USART1 Data Register Empty Hanlder 
	jmp DUMMY_IT	; USART1 TX Complete Handler 
	jmp DUMMY_IT	; Two-wire Serial Interface Handler 
	jmp DUMMY_IT	; Store Program Memory Ready Handler 

.org $0046

;****************************************************************
;* DUMMY_IT interrupt handler -- CPU hangup with LED pattern
;* (This way unhandled interrupts will be noticed)

;< többi IT kezelő a fájl végére! >

DUMMY_IT:	
	ldi r16,   0xFF ; LED pattern:  *-
	out DDRC,  r16  ;               -*
	ldi r16,   0xA5	;               *-
	out PORTC, r16  ;               -*
DUMMY_LOOP:
	rjmp DUMMY_LOOP ; endless loop

;< többi IT kezelő a fájl végére! >

;*************************************************************** 
;* MAIN program, Initialisation part
.org $004B;

RESET: 
;* Stack Pointer init, 
;  Set stack pointer to top of RAM 
	ldi temp, LOW(RAMEND) ; RAMEND = "max address in RAM"
	out SPL, temp 	      ; RAMEND value in "m128def.inc" 
	ldi temp, HIGH(RAMEND) 
	out SPH, temp 


;< ki- és bemenetek inicializálása stb > 

	; ************I/O portok inicializalasa ************

	ldi temp, 0xFF
	out DDRC, temp  ; osszes led engedelyezes kimentkent =1
	ldi temp,0x00	; a megfelelo gombokat bementkent engedelyezzuk = 0, 
    out DDRE, temp	; BT2, BT1, BT0, INT
	out DDRB, temp	; BT3 miatt kell


	;************  INT GOMB IT inicialazasa *************

	ldi  temp, 0b0000010  ; megszakítási esemény kiválasztása 
	     ;       .... 10..; pl: ISC4=10 (INT gomb lefutó élét figyeli)
	out  EICRB, temp 

	ldi  temp, 0b00010000  ; az INT gombhoz tartozó megszakítást engedélyezzük, 4 bit
	     ;       0001....  ; INT4 =1, hogy a  megszakítást engedélyezzuk
	out  EIMSK, temp 
	
	;************** POTI inicializalas ********************

	ldi  temp, 0b01100011  ; ADMUX: 5V ref, balra igazított, poti 
		 ;       01......  ; REFS = 01     (referenciafeszültség: 5V VCC) 
   	 	 ;       ..1.....  ; ADLAR = 1     (balra igazított, felso 8 bit olvasasa, 256 allap.) 
         ;       ...00011  ; ADMUX = 00011 (potméter) 
  	out  ADMUX, temp 

  	ldi  temp, 0b11100111  ; ADCSRA: folyamatos futás, IT, 128-as előosztó 
    	 ;       1.......  ; ADEN = 1   (A/D engedélyezése) 
     	 ;       .1......  ; ADSC = 1   (start conversion) 
    	 ;       ..1.....  ; ADFR = 1   (free running / folyamatos konverzió) 
    	 ;       ...0....  ; ADIF    (nem töröljük a megszakításjelző flaget) 
    	 ;       ....0...  ; ADIE = 1   (megszakítások engedélyezése) 
		 ;       .....111  ; ADPS = 111 (128-as előosztó) 
  	out  ADCSRA, temp 
	

	;************* Timer 0 inicializálása *******

	ldi tmp, 107 ; komparálandó érték (0…107 = 108)
	out OCR0, tmp
	ldi tmp, 0b00001111 ; TCCR0: CTC mód, 1024-es előosztó
			 ; 0.00.... ; FOC=0 COM=00 (kimenet tiltva)
			 ; .0..1... ; WGM=10 (CTC mód)
	 		 ; .....111 ; CS0=111 (CLK/1024)
	out TCCR0, tmp

	ldi tmp, 0b00000010 ; TIMSK: Output Compare Match IT engedélyezés
			 ; ......1. ; OCIE0=1: ha TCNT0 == OCR0, akkor IT
			 ; .......0 ; TOIE0=0 (nincs IT túlcsordulás esetén)
	out TIMSK, tmp

	sei        			; globális IT engedélyezve 
	
	call ALLIT			; a gyorsasag beallitasat vegzo rutint hivjuk
	call VELETLEN_LED	; a potenciometer zajos ledeit kihasznalva veletlen ledeket villantunk
	jmp M_LOOP


ALLIT:
	
	in  temp, ADCH    	; ADCH betöltése temp-be, poti beolvasas
	
	lsr temp			; a potenciometer erteket leosztjuk 16tal, shifteljuk a felso 8 bitet jobbra
	lsr temp			; a 16-os skalahoz kell
	lsr temp
	lsr temp
	inc temp			; megnovelljuk a temp erteket, hogy 100mstol kezdodjon a gyorsasag
	ldi tmp,0x00
	ldi r30, 0x0A		; 10et töltünk a segedregiszterbe, mert 10 szazadmasodperc lesz 0.1 sec, azaz igy jon ki a 100mseces skalazas 
	jmp Szoroz


Szoroz: 
	add tmp, r30		; hozzaadjuk a tmphez a 10-et
	dec temp 			; csokkentjuk a tempet, hogy 
	breq ALLIT2			; addig maradnuk a Szorozban, amig nem temp nem 0, Z=1-re ALLIT2
	jmp Szoroz

ALLIT2:
	mov szazad, tmp		; a megfelelo szorzoval allitjuk a TIMER-t, a szazad valtozoba belemasoljuk a az n*10-es erteket
	ret	

;*************************************************************** 
;* MAIN program, Endless loop part

M_LOOP: ;< főciklus >
	
	//in  temp, ADCH    ; ADCH betöltése temp-be, poti beolvasas

	
	in  btn, PINE   ; btn-be beolvassuk a gombokat a PINE-bol
	com  btn        ; alacsony aktív értékről magas aktívra váltjuk
	andi  btn, 0x20 ; csak BTN0 értéke kell
	lsl btn			; eltoljuk balra, a 7. bitre, mert a ledekkel ilyen sorrendben lehet konnyen osszehasonl
	lsl btn

	in btn1, PinE  
	com  btn1       ; BTN1 gomb beolvasasa
    andi btn1, 0x40	; itt nem szukseges a shifteles, mert a BT1 es a LED2, egy biten, a 6.-on vannak
	or btn, btn1 	; igy a btn-ben van a megfelelo sorrendben a BT0, BT1-es gomb erteke, 1-es bittel, ha lenyomtak
    
	in btn1, PinE   ; BTN2 gomb beolvasasa
    com  btn1 
	andi btn1, 0x80
	lsr btn1
	lsr btn1		; azert kell, hogy a BT2 a 7. bitrol a btn regiszter 5. bitjere keruljon, mert a LED1-el lesz hasonlitva
	or btn, btn1
   
    in btn1, PINB   ; BTN3 gomb beolvasasa
	com btn1;		; alacsony aktív értékről magas aktívra váltjuk
	andi btn1, 0x80	; csak a PINB7-es bit kell nekunk
    lsr btn1
	lsr btn1
	lsr btn1		; a BTN3 a 7. bitrol a 4. bitre kerul, es ezt vagyoljuk ossze a tobbi gombbal
	or  btn, btn1		; igy kialakul a btn reg. vegso sorrendje : 7:BTN0, 6: BTN1, 5: BTN2, 4:BTN3
   					
					; mivel a PORTC  3-0 bitjein van a 3:LED3, 2: LED2, 1:LED1, 0:LED0 erteke, 
					; el kell tolni a btn regisztert 4 bittel jobbra, igy a 3-0 bit hasznalhato majd osszehasonlitasra
	
	lsr btn			; hogy az also negy bitre keruljon a BTN0-3
	lsr btn
	lsr btn
	lsr btn

	ldi temp,0x00
	cp temp, btn	; megnezzuk, hogy nyomtunk-e le gombot, temp-btn mindenkepp negativ lesz, ha nyomtunk le(a bnt negalas miatt)
	brmi HASONLIT	; ha igen, N=1, a HASONLIT rutinba ugrunk
	
	jmp M_LOOP ; Endless Loop
	
	
INTKEZELES:	
	
	ldi btn,0x00	; a lenyomott gombokat tartalmazo regisztert 0-zuk
	ldi intreg, 0x00
	call ALLIT		; beallitjuk a megfelelo LEDvillanasi gyorsasagot az ALLIT rutinban
	call VELETLEN_LED; veletlen ledeket generalunk a VELETLEN_LED rutinnal
	jmp M_LOOP		; ugrunk a fo programba

HASONLIT:
	

	out PORTC, btn	; kiirodik a megnyomott gomb a PORTC 0-3 bitjére
	mov jobbled, btn; elmentjuk ezeket az ertekeket 
	lsl jobbled		; balra toljuk a lenyomott gomb ertekeket, hogy a jobboldali ledoszlopra is kikerüljön
	lsl jobbled		; PORTC 7-4 bitjeire kerul a btn erteke
	lsl jobbled
	lsl jobbled
	or jobbled, btn	; jobbled regiszterbe kerul az egesz PORTC-re kitetendo ertek
	ldi r28, 0x00	; segedvaltozo, ha nem eg a jobboldal r28=0
	cp led, btn		; osszehasonlitjuk a felvillant led es a lenyomott gomb erteket
	breq VILLOG		;
	jmp VARAKOZ

VARAKOZ: 			; ha nem villog a jobb oldal, mert nem volt talalat
	sbrc intreg,0  ; megnezzuk, hogy nyomtunk-e intet, ha 0.bit = 0, akkor maradunk itt
	jmp INTKEZELES	; egyebkent ugrunk az INTKEZELES rutinra
	jmp VARAKOZ

VILLOG:
	
	sbrc intreg,0  ; megnezzuk, hogy nyomtunk-e intet,ha 1 akkor nem csinalunk semmit
	jmp INTKEZELES ;
	sbrs r28,0 ; ha eg a jobboldali led
	jmp VILLANT
	jmp KIALSZIK

VISSZA:

	;ldi r25, 0x2A 	; decimalisan : 42 hogy 4hz legyen a villogas frekvenciaja
	ldi r25, 0xff


M_WAIT0:			; varakozo hurkok, hogy 4Hz frekvenciat allitsunk elo
	
	;ldi r26, 0xFF	;255 betoltese
	ldi r26, 0xff
M_WAIT1:	
	ldi r27, 0x06	;255 betoltese
M_WAIT2:
	dec r27
	brne M_WAIT2 ; ha nem 0 bennmaradunk az első várakozó hurokban
	dec r26
	brne M_WAIT1 ; ha nem 0 bennmaradunk a második várakozó hurokban
	dec r25
	brne M_WAIT0 ; ha nem 0 bennmaradunk a harmadik várakozó hurokban
	jmp VILLOG	 ; majd elugrunk a VILLOG rutinra 



VILLANT:
	out PORTC, jobbled  ; kitesszuk a LEDekre a lenyomott gombokat, es a talalat eseten jelzo jobboldali ledet is
	ldi r28, 0x01		; segedvaltozo, ég a jobboldali ledsor, igy r28=1

	jmp VISSZA

KIALSZIK:
	out PORTC, btn		; ahhoz kell, hogy kialudjon a jobb oldali LED oszlop, ujra csak a lenyomott gombok egnek a PORTC-n
	ldi r28,0x00		; a segedváltozó = 0, nem ég a jobboldali LED oszlop
	jmp VISSZA


VELETLEN_LED:			; veletlenszeruen LED felvillanasokat generalunk

	in veletlen, ADCL   ; beolvassuk a ket felso bitet az ADCLbol, a poti also ket helyierteku bitje
	in temp, ADCH 		; kozben a felso nyolc bitet kimentjuk, mivel letiltja az A/D konverzio utan
	rol veletlen		; eltoljuk felso ket bitet az also ket helyierteku bitre, (0-1re)
	rol veletlen		; a C miatt harom utasitas kell
	rol veletlen ; 		 
	
	ldi temp, 0x00
	cp veletlen, temp	; osszahasonlitasokkal megnezzuk a ket zajos bit erteket, es ettol fuggoen villantjuk ki a bal oldali ledoszlopot 
						; mivel minden beolvasasnal az also ket bit erteket zajos, mas-mas ertekeket generalhatunk vele
	breq LED0			; ha veletlen-temp 0, Z=1, LEDOra ugrunk, tehat veletlen=0, akkor LED0 villan fel
	
	ldi temp, 0x01		; ha veletlen erteke =1, akkor LED1 rutinnal a LED1 villan fel
	cp veletlen,temp
	breq LED1
	
	ldi temp, 0x02		; ha veletlen erteke =2, akkor LED2 rutinnal a LED2 villan fel
	cp veletlen, temp
	breq LED2
	
	jmp LED3			; ha egyikre sem ugrunk el, akkor LED3-t villantunk, az veletlen ertek ilyenkor 3


LED0:
	ldi led, 0x01		; a LED0 felvilantasa, PORTC 0. bitjet allitjuk 1-be
	out PORTC, led
	ret

LED1:
	ldi led, 0x02		; a LED1 felvilantasa, PORTC 1. bitjet allitjuk 1-be
	out PORTC, led
	ret

LED2:
	ldi led, 0x04		; a LED2 felvilantasa, PORTC 3. bitjet allitjuk 1-be
	out PORTC, led
	ret

LED3:
	ldi led, 0x08		; a LED3 felvilantasa, PORTC 4. bitjet allitjuk 1-be
	out PORTC, led
	ret

;*********** Timer IT rutin ************ 
TIMER_IT: 
 	push  temp        ; SREG-et elmentjuk a veremre, 
 	in  temp, SREG    ;
 	push  temp 
 
 	dec  szazad       ; csökkentjük a századmásodpercek számát 
	

 	brne  NEM_JART_LE ; ha nem érte el 0-t, kilépünk 
	ldi temp, 0x00
	cp temp ,btn	  ; ha nyomtunk le gombot, akkor temp-btn negativ, vagyis N=1
	brmi NEM_JART_LE	;NEM JART LE-re ugrunk ilyenkor 	
	call ALLIT			
	call VELETLEN_LED


NEM_JART_LE: 
 	pop  temp 
 	out  SREG, temp 
 	pop   temp      ; visszatöltjük SREG-et és temp-et a veremből 
 	reti 

INT_IT:				 ; INT gombhoz tartozo megszakito rutin, ha 
	push  temp        ; SREG mentése a veremre
 	in  temp, SREG    ; 
 	push  temp 

	ldi intreg, 0x01 ; allitjuk az intreg regisztert,ha lenyomtuk az int-et, intreg = 1

	pop  temp 
 	out  SREG, temp 
 	pop   temp      ; visszatöltjük SREG-et és temp-et a veremből 
 	reti

